COMITATO NAZIONALE PER L'ENERGIA NUCLEARE
Laboratori Nazionali di Frascati

LNF-62/43

M.G. Trigila-Cao: SOLUZIONE NUMERICA DI EQUAZIONI INTEGRO-DIFFERENZIALI DEL I° ORDINE DEL TIPO DI VOLTERRA, MEDIANTE CALCOLATORE I.B.M. 1620.

Nota interna: n° 140
23 Maggio 1962

Servizio Documentazione
dei Laboratori Nazionali di Frascati del CNEN
Casella Postale 70 - Frascati (Roma)

Laboratori Nazionali di Frascati del C.N.E.N.
Servizio Documentazione

LNF-62/43

Nota interna: n° 140
23 Maggio 1962

M.G. Trigila-Cao: SOLUZIONE NUMERICA DI EQUAZIONI INTEGRO-DIFFERENZIALI DEL I° ORDINE DEL TIPO DI VOLTERRA, MEDIANTE CALCOLATORE I.B.M. 1620.

Introduzione

Questo lavoro costituisce un'estensione immediata di quello apparso sulla Nota Interna n° 125 (LNF-62/22): 'Soluzione numerica di equazioni integrali di Volterra, mediante calcolatore I.B.M. 1620'.

Anche qui viene utilizzato il metodo di Runge e Kutta, noto per la sua applicazione alle equazioni differenziali ordinarie.

Fondamenti matematici

Si consideri l'equazione integro-differenziale del I° ordine del tipo di Volterra:

$$\frac{d\varphi(x)}{dx} = F[x, \varphi(x)] + \int_{x_0}^{x} G\left[x, s, \varphi(s), \left(\frac{d\varphi(x)}{dx}\right)_{x=s}\right] ds \tag{1}$$

dove le funzioni F e G sono funzioni regolari date e si conosce la $\varphi(x_0)$.

La sua soluzione numerica può essere affrontata con un metodo analogo a quello di Runge e Kutta per le equazioni differenziali ordinarie.

Infatti se si pone:

(2) $$\varphi'_\alpha(x_r) = F_r[x_r + \theta_\alpha h, \varphi_\alpha(x_r)] + h\sum_{\beta=0}^{\alpha-1} A_{\alpha\beta}\, G[x_r + \theta_\alpha h, x_r + \theta_\beta h, \varphi_\beta(x_r), \varphi'_\beta(x_r)]$$

(3) $$\varphi_\alpha(x_r) = \varphi_4(x_{r-1}) + h\sum_{\beta=0}^{\alpha-1} A_{\alpha\beta}\varphi'_\beta(x_r)$$

(4) $$\sum_{\beta=0}^{\alpha-1} A_{\alpha\beta} = \theta_\alpha$$

dove: $\varphi'(x) = \frac{d\varphi(x)}{dx}$

$r = 1,2,\ldots\ldots n-1$

n = numero di punti in cui si vuol calcolare la $\varphi(x)$

$\alpha = 1,2,3,4$

$x_r = x_o + rh$

h = passo usato durante l'elaborazione

(5) $$F_o[x,\varphi] = F[x,\varphi]$$

(6) $$\varphi'_o(x_o) = F[x_o, \varphi_o(x_o)]$$

(7) $$\varphi_o(x_o) = \varphi(x_o)$$

e per $r \geq 1$

(8) $$\varphi_o(x_r) = \varphi_4(x_{r-1})$$

(9) $$F_r[x,\varphi] = F[x,\varphi] + h\sum_{j=0}^{r-1}\sum_{\beta=0}^{3} A_{4\beta}\, G[x, x_j + \theta_\beta h, \varphi_\beta(x_j), \varphi'_\beta(x_j)]$$

e si fa in modo che gli sviluppi in serie di Taylor di $\varphi'(x_o + h) - \varphi'_4(x_o)$ e di $\varphi(x_o + h) - \varphi_4(x_o)$ comincino con il termine in h^5 si ottengono per i coefficienti $A_{\alpha\beta}$ e θ_j ($j = 0, 1,2,3,4$) le stesse relazioni che si ottengono per le equazioni differenziali ordinarie e per le equazioni integrali di Volterra.

I loro valori numerici sono riportati nella Nota Interna n° 125 (LNF-62/22).

L'errore che si commette risulta sempre minore o uguale a h^5.

Le formule che servono per la soluzione numerica della (1) sono la (2), (3), (5), (6), (7), (8), (9).

Descrizione del programma

Il programma è stato eseguito in S.P.S.

Esso è stato fatto in modo da potersi applicare facilmente a qualsiasi equazione integro-differenziale del tipo (1): basterà infatti modificare, nel modo che ora specificheremo, le istruzioni di riserva delle memorie e le istruzioni che si riferiscono al calcolo delle funzioni $F[x, \varphi]$ e $G[x, s, \varphi(s), \varphi'(s)]$.

Il programma allegato si riferisce all'equazione (b) (cfr. esempi numerici).

Le istruzioni che vanno dal label CALF al label CALG (escluso) si riferiscono al calcolo della funzione $F[x, \varphi]$.

L'argomento x si trova all'indirizzo x, in B è contenuto l'indirizzo della φ e il valore della funzione va posto all'indirizzo L.

Le istruzioni che vanno dal label CALG fino all'istruzione BB si riferiscono al calcolo della funzione $G[x, s, \varphi(s), \varphi'(s)]$.

In xB si trova il valore attuale della variabile s, in M l'indirizzo della $\varphi'(s)$ corrispondente, in N l'indirizzo della $\varphi(s)$ corrispondente. In G va posto il valore della funzione $G[x, s, \varphi(s), \varphi'(s)]$. x1 e x2 sono celle di lavoro per il calcolo delle funzioni F e G.

XO è l'estremo inferiore dell'integrale che figura nell'equazione integro-differenziale data, NU è il numero massimo di punti (diminuito di uno) in cui si vuol calcolare la $\varphi(x)$, H è il passo usato durante l'elaborazione, FIO è il valore noto di $\varphi(x_0)$.

Le istruzioni che riservano il numero delle memorie occorrenti per eseguire il calcolo (dipendente dal numero massimo di punti in cui si vuol calcolare la $\varphi(x)$) sono i DSB aventi per label FI e Z. Nel programma allegato, si è supposto che il numero NU fosse 19, quindi tenendo conto anche dei valori intermedi in cui vengono calcolate $\varphi(x)$ e $\varphi'(x)$, il numero degli elementi dei due blocchi suddetti risulta (N+2)x x 4+1 = 85. Per esempio, se si volesse calcolare $\varphi(x)$ e $\varphi'(x)$ in 50 punti, si avrebbe NU = 49, quindi queste istruzioni diverrebbero: FI DSB 10,205; Z DSB 10,205.

Nel blocco AALBE sono contenuti gli $A_{\alpha\beta}$ mentre nel blocco TETA sono contenuti i ϑ_i (i = 0,1,2,3,4) che figurano nelle formule trovate. Essi vengono memorizzati nell'ordine: A_{10}, A_{20}, A_{21}, A_{30}, A_{31}, A_{32}, A_{40}, A_{41}, A_{42}, A_{43}, ϑ_0, ϑ_1, ϑ_2, ϑ_3, ϑ_4. Questi dati devono essere introdotti da nastro, mentre XO, NU, H, FIO vanno introdotti da macchina da scrivere.

Tutti i dati, sia di ingresso che di uscita, sono in floating.

In uscita si ha: x, $\varphi(x)$, $\varphi'(x)$. Per far si che valori corrispondenti di x, $\varphi(x)$, $\varphi'(x)$ siano scritti sulla stessa riga occorre porre due Tab.: uno a 35, l'altro a 65. Dopo aver ultimato il calcolo, con una certa serie di dati di ingresso, la macchina è pronta ad accettare nuovi dati.

Esempi numerici

Il metodo è stato provato su due equazioni integro-differenziali di cui si conosce la soluzione analitica.

La prima equazione è:

(a) $$\varphi'(x) = 1+2x-\varphi(x)+\int_0^x x(1+2x)\,e^{s(x-s)}\,\varphi(s)\,ds$$

con $\varphi(0)=1$

La sua soluzione analitica è: $\varphi(x) = e^{x^2}$

La seconda equazione è:

(b) $$\varphi'(x) = e^{-x^2}(1+x^2)+x-\varphi(x)+\int_0^x x^2 e^{-sx}\varphi(s)\varphi'(s)\,ds$$

con $\varphi(0)=0$

La sua soluzione analitica è: $\varphi(x) = x$

Riportiamo nelle tabelle allegate i risultati relativi ai due casi per il confronto (cfr. Tab. I e II).

Esame dei risultati ottenuti

Il programma, come già accennato, è stato provato sia per l'equazione (a) che per quella (b) introducendo successivamente i coefficienti di Runge, Kutta e Gill.

Il calcolo è stato eseguito con otto cifre significative, usando un passo H = 0.1.

Come si vede dai risultati riportati nelle Tab. I e II, rispettivamente per l'equazione (a) e (b), l'errore che si commette usando i vari coefficienti è pressocchè lo stesso. Ciò risulta evidente specialmente nel caso dei coefficienti di Runge e di Gill.

Ringraziamenti

Desidero ringraziare vivamente il dott. A. Turrin per i preziosi consigli e l'incoraggiamento datimi durante la stesura di questo lavoro.

Bibliografia

- Symposium on the numerical treatment of ordinary differential equations, integral and integro-differential equations. (Rome, 20-24 September 1960) Pg. 363-364.

TABELLA I

Equazione (a) - Soluzione analitica $\varphi(x) = e^{x^2}$ - $\varphi(o) = 1$ - $H = 0.1$

X	e^{x^2}	$\varphi(x)$(coeff.Runge)	$\varphi(x)$(coeff.Kutta)	$\varphi(x)$(coeff.Gill)
0.1	1.0100502	1.0100509	1.0100505	1.0100509
0.2	1.0408108	1.0408124	1.0408115	1.0408124
0.3	1.0941743	1.0941772	1.0941755	1.0941772
0.4	1.1735109	1.1735155	1.1735127	1.1735155
0.5	1.2840254	1.2840325	1.2840281	1.2840325
0.6	1.4333294	1.4333401	1.4333333	1.433401
0.7	1.6323162	1.6323323	1.6323218	1.6323323
0.8	1.8964809	1.8965052	1.8964891	1.8965052
0.9	2.2479080	2.2479451	2.2479202	2.2479451
1.0	2.7182818	2.7183389	2.7183001	2.7183389

TABELLA I

Equazione (a) - Soluzione analitica $\varphi(x) = e^{x^2}$ - $\varphi(o) = 1$ - $H = 0.1$

X	$\frac{d}{dx}(e^{x^2})$	$\varphi'(x)$(coeff.Runge)	$\varphi'(x)$(coeff.Kutta)	$\varphi'(x)$(coeff.Gill)
0.1	0.20201004	0.20200924	0.20200968	0.20200924
0.2	0.41632432	0.41632240	0.41632348	0.41632240
0.3	0.65650458	0.65650091	0.65650313	0.65650092
0.4	0.93880872	0.93880250	0.93880646	0.93880251
0.5	1.2840254	1.2840150	1.2840218	1.2840150
0.6	1.7199953	1.7199786	1.7199897	1.7199785
0.7	2.2852427	2.2852166	2.2852342	2.2852167
0.8	3.0343694	3.0343287	3.0343570	3.0343288
0.9	4.0462344	4.0461708	4.0462149	4.0461708
1.0	5.4365636	5.4364637	5.4365329	5.4364632

TABELLA II

Equazione (b) - Soluzione analitica $\varphi(x) = x$ - $\varphi(o) = o$ - H = 0.1

X	$\varphi(x)$ (coeff.Runge)	$\varphi(x)$ (coeff.Kutta)	$\varphi(x)$ (coeff.Gill)	$\varphi'(x)$ (coeff.Runge)	$\varphi'(x)$ (coeff.Kutta)	$\varphi'(x)$ (coeff.Gill)
0.1	0.10000000	0.099999944	0.099999997	0.99999996	1.0000000	1.0000000
0.2	0.19999999	0.19999988	0.19999999	0.10000000	1.0000001	1.0000000
0.3	0.30000000	0.29999982	0.29999999	0.99999996	1.0000001	1.0000000
0.4	0.40000000	0.39999977	0.39999999	0.99999995	1.0000002	1.0000000
0.5	0.50000001	0.49999973	0.50000000	0.99999984	1.0000001	0.99999985
0.6	0.60000002	0.59999971	0.60000001	0.99999989	1.0000001	0.99999990
0.7	0.70000003	0.69999970	0.70000002	0.99999983	1.0000001	0.99999983
0.8	0.80000003	0.79999970	0.80000002	0.99999984	1.0000000	0.99999983
0.9	0.90000002	0.89999971	0.90000001	0.99999978	1.0000000	0.99999975
1.0	0.99999999	0.99999972	0.99999998	0.99999983	0.99999999	0.99999986
1.1	1.1099999	1.1099997	1.0999999	0.99999961	0.99999980	0.99999962
1.2	1.1999998	1.1999996	1.1999998	0.99999970	0.99999987	0.99999965
1.3	1.2999997	1.2999996	1.2999997	0.99999965	0.99999978	0.99999963
1.4	1.3999996	1.3999995	1.3999996	0.99999948	0.99999978	0.99999947
1.5	1.4999995	1.4999994	1.4999995	0.99999934	0.99999969	0.99999934
1.6	1.5999994	1.5999993	1.5999994	0.99999923	0.99999967	0.99999921
1.7	1.6999993	1.6999992	1.6999993	0.99999907	0.99999963	0.99999908
1.8	1.7999992	1.7999991	1.7999992	0.99999891	0.99999964	0.99999896
1.9	1.8999990	1.8999990	1.8999990	0.99999878	0.99999956	0.99999877
2.0	1.9999988	1.9999989	1.9999988	0.99999858	0.99999944	0.99999858

```
* EQ.INTEGRO-DIFFERENZIALE DEL PRIMO ORDINE CON IL METODO DI R.-K.
XO      DS   10
L       DS   10
FIO     DS   10
NU      DS   10
AALBE   DSB  10,10
INAAB   DSA  AALBE
TETA    DSB  10,5
INTETA  DSA  TETA
FI      DSB  10,85
INFI    DSA  FI
Z       DSB  10,85
INZ     DSA  Z
H       DS   10
A       DS   5
ALFA    DC   5,1
C       DS   5
BETA    DC   5,0
B       DS   5
LETT    DS   10
        DC   1,@
X       DS   10
        DC   1,@
R1      DC   5,0
R       DC   10,0
XR      DS   10
FR      DS   10
J       DS   10
J1      DS   5
ZERO    DC   10,0
XJ      DS   10
XBJ     DS   10
D       DS   5
M       DS   5
N       DS   5
XB      DS   10
G       DS   10
HG      DS   10
UNO     DC   10,5110000000
Q       DS   5
S1      DS   10
S2      DS   10
XRB     DS   10
K       DS   5
HS      DS   10
E       DS   5
X2      DS   10
X1      DS   10
START   RNTY XO-9
        RCTY
        RNTY NU-9
```

```
        RCTY
        RNTY H-9
        RCTY
        RNTY FIO-9
        RCTY
ALET    TF   A,ALFA
        SM   A,1
        TF   C,A
        M    A,ALFA
        SF   95
        TF   A,99
        MM   A,5
        SF   95
        TF   A,99
CIB     MM   BETA,10
        SF   95
        TF   B,99
        A    B,A
        A    B,INAAB
        RNPT LETT-9
        TF   *+18,B
        TF   ,LETT
        AM   BETA,1
        C    BETA,C
        BNH  CIB
        AM   ALFA,1
        CM   ALFA,4
        BH   LETCOM
        TFM  BETA,0
        B    ALET
LETCOM  TFM  ALFA,0
NVAL    MM   ALFA,10
        SF   95
        TF   A,99
        A    A,INTETA
        RNPT LETT-9
        TF   *+18,A
        TF   ,LETT
        AM   ALFA,1
        CM   ALFA,4
        BNH  NVAL
        TF   FI,FIO
        TF   X,XO
        TF   B,INFI
        BT   CALF,XO
        TF   Z,L
NR      TF   Q,R1
        SM   Q,1
        FM   R,H

        TF   XR,99
        FA   XR,XO
```

```
        MM    R1,40
        SF    95
        TF    D,99
        TFM   ALFA,1
NALFA   MM    ALFA,10
        SF    95
        TF    A,99
        TF    B,A
        A     B,INTETA
        TF    *+35,B
        FM    ,H

        TF    X,99
        FA    X,XR

        TF    K,ALFA
        SM    K,1
        MM    ALFA,5
        SF    95
        TF    C,99
        M     C,K
        SF    95
        TF    C,99
        TF    S2,ZERO
        TFM   BETA,0
CSUM2   MM    BETA,10
        SF    95
        TF    M,99
        TF    N,M
        A     N,D
        A     N,INZ
        A     M,C
        A     M,INAAB
        TF    *+47,N
        TF    *+47,M
        FM    ,

        TF    HS,99
        FA    S2,HS

        AM    BETA,1
        C     BETA,K
        BNH   CSUM2
        FM    H,S2

        TF    HS,99
        TF    B,D
        A     B,INFI
        TF    *+47,B
        FA    HS,
```

```
        A      B,A
        TF     E,B
        TF     *+18,B
        TF     ,HS
        BT     CALF,X
        TF     FR,L
        TFM    BETA,0
        CM     R1,1
        BN     PPUNTO
        TFM    J1,0
        TF     J,ZERO
NJ      FM     J,H

        TF     XJ,99
        FA     XJ,X0

CBETA   MM     BETA,10
        SF     95
        TF     B,99
        TF     M,B
        A      M,INTETA
        TF     *+35,M
        FM     ,H

        TF     XBJ,99
        FA     XBJ,XJ

        MM     J1,40
        SF     95
        TF     M,99
        A      M,B
        TF     N,M
        A      N,INFI
        A      M,INZ
        TF     XB,XBJ
        BT     CALG,XBJ
        FM     H,G

        TF     HG,99
        AM     B,60
        A      B,INAAB
        TF     *+35,B
        FM     ,HG

        TF     HG,99
        FA     FR,HG

        AM     BETA,1
        CM     BETA,3
        BNH    CBETA
        TFM    BETA,0
        FA     J,UNO
```

```
        AM    J1,1
        C     J1,Q
        BNH   NJ
PPUNTO  TF    S1,ZERO
CONSUM  MM    BETA,10
        SF    95
        TF    B,99
        TF    M,B
        A     M,INTETA
        TF    *+35,M
        FM    ,H

        TF    XRB,99
        FA    XRB,XR

        TF    M,B
        A     M,D
        TF    N,M
        A     N,INFI
        A     M,INZ
        TF    XB,XRB
        BT    CALG,XRB
        A     B,C
        A     B,INAAB
        TF    *+35,B
        FM    ,G

        TF    HG,99
        FA    S1,HG

        AM    BETA,1
        C     BETA,K
        BNH   CONSUM
        FM    S1,H

        TF    HG,99
        FA    HG,FR

        A     A,D
        A     A,INZ
        TF    *+18,A
        TF    ,HG
        AM    ALFA,1
        CM    ALFA,4
        BNH   NALFA
        WNTY  X-9
        TBTY
        TF    *+23,E
        TF    LETT,
        WNTY  LETT-9
        TBTY
        TF    *+23,A
```

```
        TF   LETT,
        WNTY LETT-9
        RCTY
        AM   R1,1
        FA   R,UNO

        C    R,NU
        BNH  NR
        TFM  R1,0
        TF   R,ZERO
        TFM  BETA,0
        TFM  ALFA,1
        B    START
        NOP
CALF    TF   X2,X
        FM   X2,X

        TF   X2,99
        TF   L,UNO
        FA   L,X2

        FEX  X1,X2

        FD   L,X1

        TF   L,99
        FA   L,X

        TF   *+47,B
        FS   L,

        BB
CALG    TF   X2,X
        FM   X2,X

        TF   X2,99
        FM   X,XB

        TF   X1,99
        FEX  G,X1

        FD   X2,G

        TF   G,99
        TF   *+35,M
        FM   ,G

        TF   G,99
        TF   *+35,N
        FM   ,G

        TF   G,99
        BB
        DEND START
```